

PROGRAMMA. IÁNLAT:

TV COMPUTER -
MONITOR

A program rendeltetése

A program egy gépi kód beírását megkönnyítő program, mely editor funkciókkal van ellátva. Megjegyzem, hogy ez csak egy monitor, tehát assemblerből nem fordít gépi kódra.

A program beírása

Először is írjuk be az 1. listát és indítsuk el. (Nagyon figyeljünk a csillagok számára!)

Ezután írjuk be a 2. listát és mielőtt elindítanánk, a programot másoljuk ki a:

```
POKE5895,255:SAVE"EDIMONI":  
VERIFY paranccsal.
```

Mint látjuk a programban nagyon kevés a szóköz (de, ahol szóköz van, annak ott is kell lenni, egyébként az interpreter nem érti meg), erre csak a kisebb helyfoglalás miatt van szükség.

Megjegyzés: Az 525-ös sorban a macskakörmök közé ALT+D-t kell tenni

526-osban ALT+H-t

527-esben ALT+B-t

Felhasználás

Indítsuk el a programot. Megjelenik a fejléc és a program kérdezi a kezdőcímet. Ekkor megjelenik egy lista. Elöl mindig a byte sorszáma, mögötte pedig a byte tartalma látható. A listázást ezután a joystick-kel véghezjuttatjuk fel-le, vagy új kezdő címet írhatunk be a hatványozás jel megnyomására. Középen egy kérdőjel jelenik meg, mely mindig az aktuális byte-ot mutatja. A mögötte lévő villogó "D", "H", vagy "B" azt mutatja, hogy éppen decimális, hexadecimális vagy bináris üzemmódban vagyunk. Ez azt jelenti, hogy decimálisan, hexadecimálisan vagy binárisan akarjuk az adatokat bevinni. Az üzemmódok az ALT és a D, H, B-vel válthatók. Lehet 255-nél nagyobb és lehet negatív számokat is bevinni, a program ezeket két byte-ra bontja, illetve kettes komplexumba váltja.

A parancsszavakat és hexadecimális számokat elég a kezdőbetűkkel begépelni (lehet kicsi, nagy is).

Az editor gombok közül egyedül a DEL működik. A számok vagy parancsok után értelemszerűen RETURN kell. A RETURN üres megnyomása a byte törlését jelenti.

A kezdőcímet nem lehet kisebb mint 10000 és nem lehet nagyobb mint 32767. A hibát mindig hangjelzés kíséri.

Fontos: A gépi kódú program végéről mindig tudnia kell a programnak, egyébként nem működnek az editor parancsok. Ezzel olyankor nincs gond, amikor a programot beírjuk vagy be-

töltjük. Ha viszont egy, már a tárban lévő programot akarunk editálni, akkor a monitor betöltése után a program utolsó byte-ját újból beírjuk. Ugyanis a programban van egy változó, amely mindig a legmagasabb címet jegyzi meg, és ilyen újból való beíráskor automatikusan tárolódik. Az editor parancsokhoz ez a legszükségesebb változó.

Ezt a kényelmetlenséget ellensúlyozza az, hogy a program vége akkor is megőrződik, ha meleg RESET-elünk és újra indítjuk a monitort.

```
1 ! Cím:*****HOSSZ:**Kcím:**  
2 ! SAVE *****LOAD *****INS *****DEL ****  
3 ! *****  
4 ! AAA  
10 FOR I=6685 TO 6702: READ A: POKE I,A: NEXT: ! SAVE  
20 FOR I=6708 TO 6725: READ A: POKE I,A: NEXT: ! LOAD  
30 FOR I=6730 TO 6732: READ A: POKE I,A: NEXT: ! INS  
40 FOR I=6740 TO 6742: READ A: POKE I,A: NEXT: ! DEL  
50 FOR I=6750 TO 6755: READ A: POKE I,A: NEXT: ! ATV.  
60 !  
100 DATA 17,248,25 : ! LD DE,6648  
110 DATA 247,83 : ! RST 48,83  
120 DATA 237,91,16,26: ! LD DE,(6672)  
130 DATA 237,75,9,26 : ! LD BC,(6665)  
140 DATA 247,82 : ! RST 48,82  
150 DATA 247,84 : ! RST 48,84  
160 DATA 201 : ! RET  
170 !  
180 DATA 17,248,25 : ! LD DE,6648  
190 DATA 247,211 : ! RST 48,211  
200 DATA 237,91,16,26: ! LD DE,(6672)  
210 DATA 237,75,9,26 : ! LD BC,(6665)  
220 DATA 247,210 : ! RST 48,210  
230 DATA 247,212 : ! RST 48,212  
240 DATA 201 : ! RET  
250 !  
260 DATA 237,184 : ! LDDR  
270 DATA 201 : ! RET  
280 !  
290 DATA 237,176 : ! LDIR  
300 DATA 201 : ! RET  
310 !  
320 DATA 125 : ! LD A,H  
330 DATA 18 : ! LD (DE),A  
340 DATA 19 : ! INC DE  
350 DATA 124 : ! LD A,L  
360 DATA 18 : ! LD (DE),A  
370 DATA 201 : ! RET  
400 DELETE 10-
```

1. lista

Parancsszavak

A monitor 5 parancsszót és két kiegészítő funkciót tartalmaz:

– **INS:** Hatására a program kérdezi az eltolás értékét, majd a kurzor sorában és az alatta lévő sorban lévő byte-okat az értéknek megfelelően letolja.

– **DEL:** Ugyanúgy, mint az INS, csak éppen összehúz.

– **LOAD:** Hatására a program kérdezi a betöltendő program nevét, hosszát (byte-okban) és azt, hogy hová töltsse (K, cím), majd a program betöltődik.

– **SAVE:** Értelmszerűen, mint a LOAD (K, cím=honnan). A másolás gombnyomásra indul.

– **RUN:** A gépi kódú program a „kezdőcímtől” indul.

Kiegészítők:

“\$” – ha szöveg elé írjuk, akkor az utána lévő karakterek ASCII szerint letárolódnak.

“j” – az utána következő ugrási címből kiszámolja a relatív eltolás értékét.

Működés

170–220: inicializálás + a USRTAB feltöltése

400–570: szerkesztő rész

1000–1090: a parancsszavak figyelése

1110–1140: szubrutin: egy teljes képernyős lista kiírása

1200–1255: INS szubrutin

1300–1360: DEL szubrutin

1400–1510: SAVE szubrutin

1600–1740: LOAD szubrutin

1900–2020: A számok átváltását végző szubrutin

2100–2110: “KESZ” szubrutin

2300–2370: SAVE, LOAD adatbeviteli szubrutin

2400–2480: Hexadecimális átváltó szubrutin

2500–2530: Szöveglehelyező szubrutin

2600–2630: Képernyő felfelé rotáló

2700–2730: Képernyő lefelé rotáló

2900–2910: Bináris átváltó

2920–2950: Relatív ugrás kiszámítás

Fontosabb változók:

KE: kezdőcím

EL: aktuális byte sorszám

LG: program vége

H: üzemmód mutató

Dörner Péter

A szerkesztő azért van,

hogy a lap olyan legyen,

amilyenek az olvasói!

```
170 GRAPHICS4:DIMB#*45:CH=68
180 PRINT"TV COMPUTER BASIC MONITOR":PRINT"Copyright 1986 Dörner Péter":PRINT:P
RINTSTR$(FREE);" bytes free":PRINT
210 FORI=33TO42:READ:POKEI,E:NEXT
220 DATA74,26,84,26,29,26,52,26,94,26
400 INPUTPROMPT"Kezdőcím ? ":KE
410 EL=KE:IFPEEK(6764)=0THENLG=KE+10:ELSELG=PEEK(6763)*256+PEEK(6762)
415 POKE6764,1:GRAPHICS2
420 IFEL<1E40REL>32767THENSOUND:GOTO400
440 GOSUB1110
470 B$="":PRINTAT12,1,EL:PRINTAT12,10,PEEK(EL):PRINTAT12,16,"? "
480 A$=INKEY$:PRINTCHR$(CH):CHR$(19);" ";CHR$(19)::IFA$=""THEN480
490 IFA$=CHR$(5)THENGOSUB2600:GOTO470
500 IFA$=CHR$(24)ANDEL>KE THENGOSUB2700:GOTO470
510 IFA$=CHR$(13)THENGOSUB1000:GOTO470
515 IFA$=CHR$(5)ORA$=CHR$(24)THEN480
520 IFA$="^"THENCLS:GOTO400
525 IFA$=" "THENCH=68:H=0:GOTO480
526 IFA$=" "THENCH=72:H=1:GOTO480
527 IFA$=" "THENCH=66:H=2:GOTO480
530 IFORD(A$)=8THENB$(LEN(B$))="":PRINTA$;:GOTO480
535 IFA$<" "THEN480
540 PRINTA$;
550 IFLEN(B$)=45THENSOUND:GOSUB1110:B$="":GOTO470
560 B$=B$&A$
570 GOTO480
1000 !
1010 IFB$=""THENPOKEEL,0:PRINTAT12,11,"0 " :GOSUB2600:RETURN
1015 QW$=CHR$(32ORORD(B$(1)))
1020 IFQW$="i"THENGOSUB1200:GOTO1090
1030 IFQW$="d"THENGOSUB1300:GOTO1090
1040 IFQW$="s"THENGOSUB1400:GOTO1090
1050 IFQW$="t"THENGOSUB1600:GOTO1090
1060 IFQW$="f"THENPRINTAT12,18," " :I=USR(KE):GOTO1090
1070 IFQW$="$"THENGOSUB2500:GOTO1090
1075 IFQW$="j"THENGOSUB2920:GOSUB1920:GOTO1090
1080 GOSUB1900
1090 B$="":RETURN
1110 CLS
1120 PRINT AT 2,1;
1130 FORI=EL-10TOEL+11:PRINTI;" ";PEEK(I):NEXT
1140 RETURN
1200 ! INS
1220 INPUTET:IFLG-EL+1<0THENSOUND:RETURN
1240 EXT0,LG,LG+ET,LG-EL+1
1245 FORI=EL TOEL+ET-1:POKEI,0:NEXT
1250 LG=LG+ET:EXT4,LG,6762
1255 GOSUB1110:RETURN
1300 ! DEL
1320 INPUTFH:IFLG-EL+1<0THENSOUND:RETURN
1340 EXT1,EL,EL-FH,LG-EL+1
1345 FORI=LG-FH+1TOLG:POKEI,0:NEXT
1350 LG=LG-FH:EXT4,LG,6762
1360 GOSUB1110:RETURN
1400 ! SAVE
1410 CLS
1420 GOSUB2300
1500 PRINTAT15,1,"Nyomj le egy billentyűt !"
1510 GET:EXT2:GOSUB2100:RETURN
1600 ! LOAD
1610 CLS
1620 GOSUB2300
1670 LG=KE+E-1
1740 EXT3:GOSUB2100:RETURN
1900 ! Atv.
1910 Z=VAL(B$)
1915 IFH=2THENGOSUB2900:IF0 THEN0=0:RETURN
1917 IFH=1THENGOSUB2400:IF0 THEN0=0:RETURN
1920 IFABS(Z)>32767THENSOUND:PRINTCHR$(5):GOTO2020
1940 EXT4,Z,6672
1950 IFABS(Z)>255THENPOKEEL,PEEK(6672):EL=EL+1:POKEEL,Z:PEEK(6673):GOTO1970
1960 EXT4,Z,6672:POKEEL,PEEK(6672)
1970 IFEL>LG THENLG=EL:EXT4,LG,6762
1980 IFABS(Z)>255THENGOSUB1120
1990 PRINTAT12,18," " :PRINTAT12,10,PEEK(EL)
2000 GOSUB2600
2020 RETURN
2100 PRINT:PRINT"KESZ":SOUNDPITCH4000:PRINT:PRINT"Nyomj le egy billentyűt !"
2110 GET:GOSUB1110:RETURN
2300 INPUTPROMPT"File név ? ":C$
2310 IFLEN(C$)>10THENSOUND:GOTO2300
2320 IFLEN(C$)>0THENORI=6649TO6648+LEN(C$):POKEI,ORD(C$(I-6648)):NEXT
2325 POKE6648,LEN(C$)
2330 INPUTPROMPT"K. cím ? ":KE
2340 INPUTPROMPT"Hossz ? ":E
2350 IFKE<1E40R KE>32767OR E<10R E+KE>32767THENSOUND:GOTO2330
2360 EXT4,KE,6672:EXT4,E,6665
2370 RETURN
2400 ! HEX.
2410 A=LEN(B$):Z=0
2420 FORI=1TOA
2430 IF(B$(I)>"9"ANDB$(I)<"a")ORB$(I)<"0"ORB$(I)>"f"ORLEN(B$)>5THENSOUND:Q=1:RET
URN
2440 NEXT
2450 FORI=A TO1STEP-1
2460 D=ORD(B$(I)):E=A-I
2470 IFD>96THENZ=Z+((D-87)*16^E):ELSEZ=VAL(CHR$(D)):Z=Z+(D*16^E)
2480 NEXT:RETURN
2500 K=LEN(B$):IFK<2THENSOUND:RETURN
2510 FORI=2TOK:POKEEL,ORD(B$(I)):EL=EL+1:NEXT
2520 IFEL>LG THENLG=EL
2530 EL=EL-K+1:GOSUB1110:RETURN
2600 ! Fe1
2610 PRINTAT12,16," " :PRINTAT1,1,CHR$(25):EL=EL+1
2620 PRINTAT23,1,EL+11:PRINTAT23,10,PEEK(EL+11)
2630 RETURN
2700 ! Le
2710 PRINTAT12,16," " :PRINTAT1,1,CHR$(14):EL=EL-1
2720 PRINTAT2,1,EL-10:PRINTAT2,10,PEEK(EL-10)
2730 RETURN
2900 IFLEN(B$)<8THENSOUND:Q=1:RETURN
2910 Z=0:FORI=0TO7:D=VAL(B$(8-I))*2^I:Z=Z+D:NEXT:RETURN
2920 IFLEN(B$)<2THENSOUND:RETURN:ELSEB$=B$(2)
2930 I=VAL(B$)-EL-1
2940 IFI<-127ORI>128THENSOUND:RETURN
2950 Z=I:RETURN
```

2. lista